Rules for integrands involving exponential integral functions

1. $\int u \, \text{ExpIntegralE}[n, a + b x] \, dx$

   1: $\int \text{ExpIntegralE}[n, a + b x] \, dx$

   Basis: $\frac{\partial E_n(z)}{\partial z} = -E_{n-1}(z)$

   Rule:

$$\int \text{ExpIntegralE}[n, a + b x] \, dx \longrightarrow -\frac{\text{ExpIntegralE}[n + 1, a + b x]}{b}$$

   Program code:

```
Int[ExpIntegralE[n_,a_.+b_.*x_],x_Symbol] :=
  -ExpIntegralE[n+1,a+b*x]/b /;
FreeQ[{a,b,n},x]
```

2. $\int (d\,x)^m\,\text{ExpIntegralE}[n,\,b\,x]\,dx$

   1. $\int (d\,x)^m\,\text{ExpIntegralE}[n,\,b\,x]\,dx$ when $m+n==0$

      1. $\int x^m\,\text{ExpIntegralE}[n,\,b\,x]\,dx$ when $m+n==0\,\wedge\,m\in\mathbb{Z}$

         1: $\int x^m\,\text{ExpIntegralE}[n,\,b\,x]\,dx$ when $m+n==0\,\wedge\,m\in\mathbb{Z}^+$

Derivation: Inverted integration by parts

Rule: If $m+n==0\,\wedge\,m\in\mathbb{Z}^+$, then

$$\int x^m\,\text{ExpIntegralE}[n,\,b\,x]\,dx\;\longrightarrow\;-\frac{x^m\,\text{ExpIntegralE}[n+1,\,b\,x]}{b}+\frac{m}{b}\int x^{m-1}\,\text{ExpIntegralE}[n+1,\,b\,x]\,dx$$

Program code:

```
Int[x_^m_.*ExpIntegralE[n_,b_.*x_],x_Symbol] :=
  -x^m*ExpIntegralE[n+1,b*x]/b +
  m/b*Int[x^(m-1)*ExpIntegralE[n+1,b*x],x] /;
FreeQ[b,x] && EqQ[m+n,0] && IGtQ[m,0]
```

2. $\int x^m \text{ExpIntegralE}[n, b x] \, dx$ when $m + n == 0 \land m \in \mathbb{Z}^-$

1: $\int \frac{\text{ExpIntegralE}[1, b x]}{x} \, dx$

Rule:

$$\int \frac{\text{ExpIntegralE}[1, b x]}{x} \, dx \longrightarrow b\, x\, \text{HypergeometricPFQ}[\{1, 1, 1\}, \{2, 2, 2\}, -b x] - \text{EulerGamma}\, \text{Log}[x] - \frac{1}{2} \text{Log}[b x]^2$$

Program code:

```
Int[ExpIntegralE[1,b_.*x_]/x_,x_Symbol] :=
  b*x*HypergeometricPFQ[{1,1,1},{2,2,2},-b*x] - EulerGamma*Log[x] - 1/2*Log[b*x]^2 /;
FreeQ[b,x]
```

2: $\int x^m \text{ExpIntegralE}[n, b x] \, dx$ when $m + n == 0 \land m + 1 \in \mathbb{Z}^-$

Derivation: Integration by parts

Rule: If $m + n == 0 \land m + 1 \in \mathbb{Z}^-$, then

$$\int x^m \text{ExpIntegralE}[n, b x] \, dx \longrightarrow \frac{x^{m+1} \text{ExpIntegralE}[n, b x]}{m + 1} + \frac{b}{m + 1} \int x^{m+1} \text{ExpIntegralE}[n - 1, b x] \, dx$$

Program code:

```
Int[x_^m_*ExpIntegralE[n_,b_.*x_],x_Symbol] :=
  x^(m+1)*ExpIntegralE[n,b*x]/(m+1) +
  b/(m+1)*Int[x^(m+1)*ExpIntegralE[n-1,b*x],x] /;
FreeQ[b,x] && EqQ[m+n,0] && ILtQ[m,-1]
```

2: $\int (d\,x)^m\,\text{ExpIntegralE}[n,\,b\,x]\,dx$ when $m+n==0\,\land\,m\notin\mathbb{Z}$

Rule: If $m+n==0\,\land\,m\notin\mathbb{Z}$, then

$$\int (d\,x)^m\,\text{ExpIntegralE}[n,\,b\,x]\,dx\,\longrightarrow\,\frac{(d\,x)^m\,\text{Gamma}[m+1]\,\text{Log}[x]}{b\,(b\,x)^m}-\frac{(d\,x)^{m+1}\,\text{HypergeometricPFQ}[\{m+1,\,m+1\},\,\{m+2,\,m+2\},\,-b\,x]}{d\,(m+1)^2}$$

Program code:

```
Int[(d_.*x_)^m_*ExpIntegralE[n_,b_.*x_],x_Symbol] :=
  (d*x)^m*Gamma[m+1]*Log[x]/(b*(b*x)^m) - (d*x)^(m+1)*HypergeometricPFQ[{m+1,m+1},{m+2,m+2},-b*x]/(d*(m+1)^2) /;
FreeQ[{b,d,m,n},x] && EqQ[m+n,0] && Not[IntegerQ[m]]
```

2: $\int (d\,x)^m\,\text{ExpIntegralE}[n,\,b\,x]\,dx$ when $m+n\neq 0$

Rule: If $m+n\neq 0$, then

$$\int (d\,x)^m\,\text{ExpIntegralE}[n,\,b\,x]\,dx\,\longrightarrow\,\frac{(d\,x)^{m+1}\,\text{ExpIntegralE}[n,\,b\,x]}{d\,(m+n)}-\frac{(d\,x)^{m+1}\,\text{ExpIntegralE}[-m,\,b\,x]}{d\,(m+n)}$$

Program code:

```
Int[(d_.*x_)^m_.*ExpIntegralE[n_,b_.*x_],x_Symbol] :=
  (d*x)^(m+1)*ExpIntegralE[n,b*x]/(d*(m+n)) - (d*x)^(m+1)*ExpIntegralE[-m,b*x]/(d*(m+n)) /;
FreeQ[{b,d,m,n},x] && NeQ[m+n,0]
```

3. $\int (c+d\,x)^m\,\text{ExpIntegralE}[n,\,a+b\,x]\,dx$

1: $\int (c+d\,x)^m\,\text{ExpIntegralE}[n,\,a+b\,x]\,dx$ when $m\in\mathbb{Z}^+\lor n\in\mathbb{Z}^-\lor(m>0\land n<-1)$

Derivation: Inverted integration by parts

Rule: If $m\in\mathbb{Z}^+\lor n\in\mathbb{Z}^-\lor(m>0\land n<-1)$, then

$$\int (c+d\,x)^m\,\text{ExpIntegralE}[n,\,a+b\,x]\,dx \longrightarrow -\frac{(c+d\,x)^m\,\text{ExpIntegralE}[n+1,\,a+b\,x]}{b}+\frac{d\,m}{b}\int (c+d\,x)^{m-1}\,\text{ExpIntegralE}[n+1,\,a+b\,x]\,dx$$

Program code:

```
Int[(c_.+d_.*x_)^m_.*ExpIntegralE[n_,a_+b_.*x_],x_Symbol] :=
  -(c+d*x)^m*ExpIntegralE[n+1,a+b*x]/b +
  d*m/b*Int[(c+d*x)^(m-1)*ExpIntegralE[n+1,a+b*x],x] /;
FreeQ[{a,b,c,d,m,n},x] && (IGtQ[m,0] || ILtQ[n,0] || GtQ[m,0] && LtQ[n,-1])
```

2: $\int (c+d\,x)^m\,\text{ExpIntegralE}[n,\,a+b\,x]\,dx$ when $(n\in\mathbb{Z}^+\lor(m<-1\land n>0))\land m\neq-1$

Derivation: Integration by parts

Rule: If $(n\in\mathbb{Z}^+\lor(m<-1\land n>0))\land m\neq-1$, then

$$\int (c+d\,x)^m\,\text{ExpIntegralE}[n,\,a+b\,x]\,dx \longrightarrow \frac{(c+d\,x)^{m+1}\,\text{ExpIntegralE}[n,\,a+b\,x]}{d\,(m+1)}+\frac{b}{d\,(m+1)}\int (c+d\,x)^{m+1}\,\text{ExpIntegralE}[n-1,\,a+b\,x]\,dx$$

Program code:

```
Int[(c_.+d_.*x_)^m_.*ExpIntegralE[n_,a_+b_.*x_],x_Symbol] :=
  (c+d*x)^(m+1)*ExpIntegralE[n,a+b*x]/(d*(m+1)) +
  b/(d*(m+1))*Int[(c+d*x)^(m+1)*ExpIntegralE[n-1,a+b*x],x] /;
FreeQ[{a,b,c,d,m,n},x] && (IGtQ[n,0] || LtQ[m,-1] && GtQ[n,0]) && NeQ[m,-1]
```

3: $\int (c + d\,x)^m \,\text{ExpIntegralE}[n,\ a + b\,x]\, dx$

Rule:

$$\int (c + d\,x)^m \,\text{ExpIntegralE}[n,\ a + b\,x]\, dx \longrightarrow \int (c + d\,x)^m \,\text{ExpIntegralE}[n,\ a + b\,x]\, dx$$

Program code:

```
Int[(c_.+d_.*x_)^m_.*ExpIntegralE[n_,a_+b_.*x_],x_Symbol] :=
  Unintegrable[(c+d*x)^m*ExpIntegralE[n,a+b*x],x] /;
FreeQ[{a,b,c,d,m,n},x]
```

2. $\int u\, \text{ExpIntegralEi}[a + b\,x]\, dx$

1: $\int \text{ExpIntegralEi}[a + b\,x]\, dx$

Derivation: Integration by parts

Rule:

$$\int \text{ExpIntegralEi}[a + b\,x]\, dx \longrightarrow \frac{(a + b\,x)\ \text{ExpIntegralEi}[a + b\,x]}{b} - \frac{e^{a+b\,x}}{b}$$

Program code:

```
Int[ExpIntegralEi[a_.+b_.*x_],x_Symbol] :=
  (a+b*x)*ExpIntegralEi[a+b*x]/b - E^(a+b*x)/b /;
FreeQ[{a,b},x]
```

2. $\int (c + d x)^m \text{ExpIntegralEi}[a + b x] \, dx$

1. $\int \frac{\text{ExpIntegralEi}[a + b x]}{c + d x} \, dx$

1: $\int \frac{\text{ExpIntegralEi}[b x]}{x} \, dx$

Derivation: Piecewise constant extraction

Basis: $\partial_x (\text{ExpIntegralEi}[b x] + \text{ExpIntegralE}[1, -b x]) == 0$

Rule:

$$\int \frac{\text{ExpIntegralEi}[b x]}{x} \, dx \longrightarrow (\text{ExpIntegralEi}[b x] + \text{ExpIntegralE}[1, -b x]) \int \frac{1}{x} \, dx - \int \frac{\text{ExpIntegralE}[1, -b x]}{x} \, dx$$

$$\longrightarrow \text{Log}[x] \, (\text{ExpIntegralEi}[b x] + \text{ExpIntegralE}[1, -b x]) - \int \frac{\text{ExpIntegralE}[1, -b x]}{x} \, dx$$

Program code:

```
Int[ExpIntegralEi[b_.*x_]/x_,x_Symbol] :=
  Log[x]*(ExpIntegralEi[b*x]+ExpIntegralE[1,-b*x]) - Int[ExpIntegralE[1,-b*x]/x,x] /;
FreeQ[b,x]
```

X: $\int \frac{\text{ExpIntegralEi}[a+b\,x]}{c+d\,x}\,dx$

Rule:

$$\int \frac{\text{ExpIntegralEi}[a+b\,x]}{c+d\,x}\,dx \longrightarrow \int \frac{\text{ExpIntegralEi}[a+b\,x]}{c+d\,x}\,dx$$

Program code:

```
Int[ExpIntegralEi[a_.+b_.*x_]/(c_.+d_.*x_),x_Symbol] :=
  Unintegrable[ExpIntegralEi[a+b*x]/(c+d*x),x] /;
FreeQ[{a,b,c,d},x]
```

2: $\int (c+d\,x)^m\,\text{ExpIntegralEi}[a+b\,x]\,dx$ when $m \neq -1$

Derivation: Integration by parts

Rule: If $m \neq -1$, then

$$\int (c+d\,x)^m\,\text{ExpIntegralEi}[a+b\,x]\,dx \longrightarrow \frac{(c+d\,x)^{m+1}\,\text{ExpIntegralEi}[a+b\,x]}{d\,(m+1)} - \frac{b}{d\,(m+1)}\int \frac{(c+d\,x)^{m+1}\,e^{a+b\,x}}{a+b\,x}\,dx$$

Program code:

```
Int[(c_.+d_.*x_)^m_.*ExpIntegralEi[a_.+b_.*x_],x_Symbol] :=
  (c+d*x)^(m+1)*ExpIntegralEi[a+b*x]/(d*(m+1)) -
  b/(d*(m+1))*Int[(c+d*x)^(m+1)*E^(a+b*x)/(a+b*x),x] /;
FreeQ[{a,b,c,d,m},x] && NeQ[m,-1]
```

3. $\int u\,\text{ExpIntegralEi}[a+b\,x]^2\,dx$

1: $\int \text{ExpIntegralEi}[a+b\,x]^2\,dx$

Derivation: Integration by parts

Rule:

$$\int \text{ExpIntegralEi}[a+b\,x]^2\,dx \longrightarrow \frac{(a+b\,x)\ \text{ExpIntegralEi}[a+b\,x]^2}{b} - 2\int e^{a+b\,x}\,\text{ExpIntegralEi}[a+b\,x]\,dx$$

Program code:

```
Int[ExpIntegralEi[a_.+b_.*x_]^2,x_Symbol] :=
  (a+b*x)*ExpIntegralEi[a+b*x]^2/b -
  2*Int[E^(a+b*x)*ExpIntegralEi[a+b*x],x] /;
FreeQ[{a,b},x]
```

2. $\int x^m \text{ExpIntegralEi}[a+b\,x]^2\,dx$

1: $\int x^m \text{ExpIntegralEi}[b\,x]^2\,dx$ when $m\in\mathbb{Z}^+$

Derivation: Integration by parts

Rule: If $m\in\mathbb{Z}^+$, then

$$\int x^m \text{ExpIntegralEi}[b\,x]^2\,dx \longrightarrow \frac{x^{m+1}\text{ExpIntegralEi}[b\,x]^2}{m+1}-\frac{2}{m+1}\int x^m e^{b\,x}\text{ExpIntegralEi}[b\,x]\,dx$$

Program code:

```
Int[x_^m_.*ExpIntegralEi[b_.*x_]^2,x_Symbol] :=
  x^(m+1)*ExpIntegralEi[b*x]^2/(m+1) -
  2/(m+1)*Int[x^m*E^(b*x)*ExpIntegralEi[b*x],x] /;
FreeQ[b,x] && IGtQ[m,0]
```

2: $\int x^m \text{ExpIntegralEi}[a+b\,x]^2\,dx$ when $m\in\mathbb{Z}^+$

Derivation: Iterated integration by parts

Rule: If $m\in\mathbb{Z}^+$, then

$$\int x^m \text{ExpIntegralEi}[a+b\,x]^2\,dx \longrightarrow$$

$$\frac{x^{m+1}\,\text{ExpIntegralEi}[a+b\,x]^2}{m+1}+\frac{a\,x^m\,\text{ExpIntegralEi}[a+b\,x]^2}{b\,(m+1)}-\frac{2}{m+1}\int x^m\,e^{a+b\,x}\,\text{ExpIntegralEi}[a+b\,x]\,dx-\frac{a\,m}{b\,(m+1)}\int x^{m-1}\,\text{ExpIntegralEi}[a+b\,x]^2\,dx$$

Program code:

```
Int[x_^m_.*ExpIntegralEi[a_+b_.*x_]^2,x_Symbol] :=
  x^(m+1)*ExpIntegralEi[a+b*x]^2/(m+1) +
  a*x^m*ExpIntegralEi[a+b*x]^2/(b*(m+1)) -
  2/(m+1)*Int[x^m*E^(a+b*x)*ExpIntegralEi[a+b*x],x] -
  a*m/(b*(m+1))*Int[x^(m-1)*ExpIntegralEi[a+b*x]^2,x] /;
FreeQ[{a,b},x] && IGtQ[m,0]
```

x: $\int x^m \, \text{ExpIntegralEi}[a + b\, x]^2 \, dx$ when $m + 2 \in \mathbb{Z}^-$

Derivation: Inverted integration by parts

Rule: If $m + 2 \in \mathbb{Z}^-$, then

$$\int x^m \, \text{ExpIntegralEi}[a + b\, x]^2 \, dx \longrightarrow$$

$$\frac{b\, x^{m+2}\, \text{ExpIntegralEi}[a + b\, x]^2}{a\, (m+1)} + \frac{x^{m+1}\, \text{ExpIntegralEi}[a + b\, x]^2}{m+1} - \frac{2\, b}{a\, (m+1)} \int x^{m+1}\, e^{a+b\, x}\, \text{ExpIntegralEi}[a + b\, x] \, dx - \frac{b\, (m+2)}{a\, (m+1)} \int x^{m+1}\, \text{ExpIntegralEi}[a + b\, x]^2 \, dx$$

Program code:

```
(* Int[x_^m_.*ExpIntegralEi[a_+b_.*x_]^2,x_Symbol] :=
  b*x^(m+2)*ExpIntegralEi[a+b*x]^2/(a*(m+1)) +
  x^(m+1)*ExpIntegralEi[a+b*x]^2/(m+1) -
  2*b/(a*(m+1))*Int[x^(m+1)*E^(a+b*x)*ExpIntegralEi[a+b*x],x] -
  b*(m+2)/(a*(m+1))*Int[x^(m+1)*ExpIntegralEi[a+b*x]^2,x] /;
FreeQ[{a,b},x] && ILtQ[m,-2] *)
```

4. $\int u\, e^{a+b x}\, \text{ExpIntegralEi}[c+d x]\, dx$

1: $\int e^{a+b x}\, \text{ExpIntegralEi}[c+d x]\, dx$

Derivation: Integration by parts

– Rule:

$$\int e^{a+b x}\, \text{ExpIntegralEi}[c+d x]\, dx \longrightarrow \frac{e^{a+b x}\, \text{ExpIntegralEi}[c+d x]}{b}-\frac{d}{b}\int \frac{e^{a+c+(b+d) x}}{c+d x}\, dx$$

– Program code:

```
Int[E^(a_.+b_.*x_)*ExpIntegralEi[c_.+d_.*x_],x_Symbol] :=
  E^(a+b*x)*ExpIntegralEi[c+d*x]/b -
  d/b*Int[E^(a+c+(b+d)*x)/(c+d*x),x] /;
FreeQ[{a,b,c,d},x]
```

2. $\int x^m e^{a+b x}\,\mathtt{ExpIntegralEi}[c+d x]\,dx$

1: $\int x^m e^{a+b x}\,\mathtt{ExpIntegralEi}[c+d x]\,dx$ when $m \in \mathbb{Z}^+$

Derivation: Integration by parts

Rule: If $m \in \mathbb{Z}^+$, then

$$\int x^m e^{a+b x}\,\mathtt{ExpIntegralEi}[c+d x]\,dx \longrightarrow$$

$$\frac{x^m e^{a+b x}\,\mathtt{ExpIntegralEi}[c+d x]}{b}-\frac{d}{b}\int \frac{x^m e^{a+c+(b+d) x}}{c+d x}\,dx-\frac{m}{b}\int x^{m-1} e^{a+b x}\,\mathtt{ExpIntegralEi}[c+d x]\,dx$$

Program code:

```
Int[x_^m_.*E^(a_.+b_.*x_)*ExpIntegralEi[c_.+d_.*x_],x_Symbol] :=
  x^m*E^(a+b*x)*ExpIntegralEi[c+d*x]/b -
  d/b*Int[x^m*E^(a+c+(b+d)*x)/(c+d*x),x] -
  m/b*Int[x^(m-1)*E^(a+b*x)*ExpIntegralEi[c+d*x],x] /;
FreeQ[{a,b,c,d},x] && IGtQ[m,0]
```

2: $\int x^m e^{a+b x} \text{ExpIntegralEi}[c+d x] \, dx$ when $m+1 \in \mathbb{Z}^-$

Derivation: Inverted integration by parts

Rule: If $m+1 \in \mathbb{Z}^-$, then

$$\int x^m e^{a+b x} \text{ExpIntegralEi}[c+d x] \, dx \longrightarrow$$

$$\frac{x^{m+1} e^{a+b x} \text{ExpIntegralEi}[c+d x]}{m+1} - \frac{d}{m+1} \int \frac{x^{m+1} e^{a+c+(b+d) x}}{c+d x} \, dx - \frac{b}{m+1} \int x^{m+1} e^{a+b x} \text{ExpIntegralEi}[c+d x] \, dx$$

Program code:

```
Int[x_^m_*E^(a_.+b_.*x_)*ExpIntegralEi[c_.+d_.*x_],x_Symbol] :=
  x^(m+1)*E^(a+b*x)*ExpIntegralEi[c+d*x]/(m+1) -
  d/(m+1)*Int[x^(m+1)*E^(a+c+(b+d)*x)/(c+d*x),x] -
  b/(m+1)*Int[x^(m+1)*E^(a+b*x)*ExpIntegralEi[c+d*x],x] /;
FreeQ[{a,b,c,d},x] && ILtQ[m,-1]
```

5. $\int u\,\text{ExpIntegralEi}\left[d\,\left(a+b\,\text{Log}\left[c\,x^n\right]\right)\right]\,dx$

1: $\int \text{ExpIntegralEi}\left[d\,\left(a+b\,\text{Log}\left[c\,x^n\right]\right)\right]\,dx$

Derivation: Integration by parts

Basis: $\partial_x\,\text{ExpIntegralEi}\left[d\,\left(a+b\,\text{Log}\left[c\,x^n\right]\right)\right] == \frac{b\,n\,e^{a\,d}\,\left(c\,x^n\right)^{b\,d}}{x\,\left(a+b\,\text{Log}\left[c\,x^n\right]\right)}$

Rule: If $m \neq -1$, then

$$\int \text{ExpIntegralEi}\left[d\,\left(a+b\,\text{Log}\left[c\,x^n\right]\right)\right]\,dx \longrightarrow x\,\text{ExpIntegralEi}\left[d\,\left(a+b\,\text{Log}\left[c\,x^n\right]\right)\right]-b\,n\,e^{a\,d}\int \frac{\left(c\,x^n\right)^{b\,d}}{a+b\,\text{Log}\left[c\,x^n\right]}\,dx$$

–

Program code:

```
Int[ExpIntegralEi[d_.*(a_.+b_.*Log[c_.*x_^n_.])],x_Symbol] :=
  x*ExpIntegralEi[d*(a+b*Log[c*x^n])] - b*n*E^(a*d)*Int[(c*x^n)^(b*d)/(a+b*Log[c*x^n]),x] /;
FreeQ[{a,b,c,d,n},x]
```

2: $\int \frac{\text{ExpIntegralEi}\left[d\,\left(a+b\,\text{Log}\left[c\,x^n\right]\right)\right]}{x}\,dx$

Derivation: Integration by substitution

Basis: $\frac{F[\text{Log}[c\,x^n]]}{x} == \frac{1}{n}\,\text{Subst}[F[x],\,x,\,\text{Log}[c\,x^n]]\;\partial_x \text{Log}[c\,x^n]$

Rule:

$$\int \frac{\text{ExpIntegralEi}\left[d\,\left(a+b\,\text{Log}\left[c\,x^n\right]\right)\right]}{x}\,dx \longrightarrow \frac{1}{n}\,\text{Subst}\left[\text{ExpIntegralEi}[d\,(a+b\,x)],\,x,\,\text{Log}\left[c\,x^n\right]\right]$$

Program code:

```
Int[ExpIntegralEi[d_.*(a_.+b_.*Log[c_.*x_^n_.])]/x_,x_Symbol] :=
  1/n*Subst[ExpIntegralEi[d*(a+b*x)],x,Log[c*x^n]] /;
FreeQ[{a,b,c,d,n},x]
```

3: $\int (e\,x)^m\, \text{ExpIntegralEi}\left[d\,\left(a+b\,\text{Log}\left[c\,x^n\right]\right)\right] dx$ when $m \neq -1$

Derivation: Integration by parts

Basis: $\partial_x \text{ExpIntegralEi}[d\,(a+b\,\text{Log}[c\,x^n])] == \frac{b\,n\,e^{a\,d}\,(c\,x^n)^{b\,d}}{x\,(a+b\,\text{Log}[c\,x^n])}$

Rule: If $m \neq -1$, then

$$\int (e\,x)^m\, \text{ExpIntegralEi}\left[d\,\left(a+b\,\text{Log}\left[c\,x^n\right]\right)\right] dx \longrightarrow \frac{(e\,x)^{m+1}\, \text{ExpIntegralEi}\left[d\,\left(a+b\,\text{Log}\left[c\,x^n\right]\right)\right]}{e\,(m+1)} - \frac{b\,n\,e^{a\,d}\,\left(c\,x^n\right)^{b\,d}}{(m+1)\,(e\,x)^{b\,d\,n}} \int \frac{(e\,x)^{m+b\,d\,n}}{a+b\,\text{Log}\left[c\,x^n\right]}\, dx$$

Program code:

```
Int[(e_.*x_)^m_.*ExpIntegralEi[d_.*(a_.+b_.*Log[c_.*x_^n_.])],x_Symbol] :=
  (e*x)^(m+1)*ExpIntegralEi[d*(a+b*Log[c*x^n])]/(e*(m+1)) -
  b*n*E^(a*d)*(c*x^n)^(b*d)/((m+1)*(e*x)^(b*d*n))*Int[(e*x)^(m+b*d*n)/(a+b*Log[c*x^n]),x] /;
FreeQ[{a,b,c,d,e,m,n},x] && NeQ[m,-1]
```

Rules for integrands involving logarithmic integral functions

1: $\int \text{LogIntegral}[a + b\,x]\,dx$

Derivation: Integration by parts

Rule:

$$\int \text{LogIntegral}[a + b\,x]\,dx \longrightarrow \frac{(a + b\,x)\,\text{LogIntegral}[a + b\,x]}{b} - \frac{\text{ExpIntegralEi}[2\,\text{Log}[a + b\,x]]}{b}$$

Program code:

```
Int[LogIntegral[a_.+b_.*x_],x_Symbol] :=
  (a+b*x)*LogIntegral[a+b*x]/b - ExpIntegralEi[2*Log[a+b*x]]/b /;
FreeQ[{a,b},x]
```

2. $\int (c + d\,x)^m\,\text{LogIntegral}[a + b\,x]\,dx$

1. $\int \frac{\text{LogIntegral}[a + b\,x]}{c + d\,x}\,dx$

1: $\int \frac{\text{LogIntegral}[b\,x]}{x}\,dx$

Rule:

$$\int \frac{\text{LogIntegral}[b\,x]}{x}\,dx \longrightarrow -b\,x + \text{Log}[b\,x]\,\text{LogIntegral}[b\,x]$$

Program code:

```
Int[LogIntegral[b_.*x_]/x_,x_Symbol] :=
  -b*x + Log[b*x]*LogIntegral[b*x] /;
FreeQ[b,x]
```

U: $\int \frac{\text{LogIntegral}[a+b\,x]}{c+d\,x}\,dx$

Rule:

$$\int \frac{\text{LogIntegral}[a+b\,x]}{c+d\,x}\,dx \longrightarrow \int \frac{\text{LogIntegral}[a+b\,x]}{c+d\,x}\,dx$$

Program code:

```
Int[LogIntegral[a_.+b_.*x_]/(c_.+d_.*x_),x_Symbol] :=
  Unintegrable[LogIntegral[a+b*x]/(c+d*x),x] /;
FreeQ[{a,b,c,d},x]
```

2: $\int (c+d\,x)^m\,\text{LogIntegral}[a+b\,x]\,dx$ when $m \neq -1$

Derivation: Integration by parts

Rule: If $m \neq -1$, then

$$\int (c+d\,x)^m\,\text{LogIntegral}[a+b\,x]\,dx \longrightarrow \frac{(c+d\,x)^{m+1}\,\text{LogIntegral}[a+b\,x]}{d\,(m+1)} - \frac{b}{d\,(m+1)}\int \frac{(c+d\,x)^{m+1}}{\text{Log}[a+b\,x]}\,dx$$

Program code:

```
Int[(c_.+d_.*x_)^m_.*LogIntegral[a_.+b_.*x_],x_Symbol] :=
  (c+d*x)^(m+1)*LogIntegral[a+b*x]/(d*(m+1)) - b/(d*(m+1))*Int[(c+d*x)^(m+1)/Log[a+b*x],x] /;
FreeQ[{a,b,c,d,m},x] && NeQ[m,-1]
```